
=========State_Variable_s_eq_1================
STATE VARIABLE FILTERS MAP POLES AND ZEROS DIRECTLY
TO CIRCUIT COMPONENT VALUES.  

State_Variable_s_eq_1
 *   dsauersanjose@aol.com 
 *   www.idea2ic.com                               R3 = 1/A0
 *   4/15/08             ________________________/\  /\  /\__
 *                      |                          \/  \/    |
 *                      |               R2 = 1/A1            |
 *                      |____________/\  /\  /\_             |
 *                      |              \/  \/   |            |
 *                      |   R1 = 1              |BP          |LP
 *                      |_/\  /\  /\_ HP        |            |
 *          R0 = 1/A0   |   \/  \/   |          |            |
 *     ___              | |\         | XONE_S1  |  XONE_S2   |
 *    |IN |_/\  /\  /\__|_|-\        | _______  |  _______   |  ___
 *    |___|   \/  \/  SUMN|  \_______|_|      |_|__|      |__|_|OUT|
 *     _|_                |  /         | 1/s  |    | 1/s  |    |___|
 *    /VIN\             __|+/ OPA1     |______|    |______|
 *    \___/            |  |/
 *      |             _|_        H(s) = 1/(s^2 +s*A1   + A0)
 *     _|_            ///        H(s) = 1/(s^2 +s*w0/Q + w0^2)
 *     ///
 *             Set  A0 = 1 and w0^2  = s^2  = 1 
 *             Then R2 = Q and  2*PI*0.159Hz = 1



.OPTIONS  GMIN=1e-18    METHOD=trap   srcsteps = 1  gminsteps = 1
*======== ====== ====== ====== ====== ====== ====== ====== ======
V_IN      VIN    0      AC     1      DC     0
R0        VIN    SUMN   1    
R1        SUMN   HP     1
R2        SUMN   BP     10
R3        SUMN   LP     1
XOPA1     SUMN   0      HP    OPA
XONES1    HP     BP           ONE_S
XONES2    BP     LP           ONE_S
.ac       dec    50     .01   1

*==State_Variable_Filters_Map_Equations_To_Components====

.control
run
plot   db(bp)  db(hp) db(lp)   title   StateVariable_Q_10
*==========FeedBack_Adjusts_Q============================
alter  R2     resistance =    1
run
plot  db(bp)  db(hp) db(lp)   title   StateVariable_Q_1
.endc
*
*             C1 1F                ___
*             _||______ O1   _____|OUT|
*  ___    R1 | || _|_  |    _|_   |___|     .SUBCKT   ONE_S  IN   OUT
* |IN |_/\  _|__ /   \ |__ /   \            ___  _______    ___
* |___|   \/  IP| Egn1|   | Egn2|   ---->  |IN |_|      |__|OUT|
*       1      _| 90dB|  _|  -1 |   ---->  |___| | 1/s  |  |___|
*             |  \___/  |  \___/                 |______|
*             |____|    |____|            
*                 _|_       _|_            s  = w0     s = 2*PI*Freq
*                 ///       ///            @s = 1      R = 1/sC 
*
*
*   http://www.idea2ic.com/PlayWithJavascript/R_C_Freq.html
*
*    ___         O1         C1 10pF   ___     .SUBCKT  OPA INP INN OUT
*   |INP|_     _____        _||______|OUT|    
*   |___| |   _|_   |   R1 | || _|_  |___|     ___   |\
*         |_ /   \  |_/\  _|__ /   \          |INP|__|-\    ___
*    ___    | EGN1|  10K\/  IP| EGN2|  ---->  |___|  |  \__|OUT|
*   |INN|___|  -1 |          _| 90dB|  ---->   ___   |  /  |___|
*   |___|    \___/       IP |  \___/          |INN|__|+/
*              |            |____|            |___|  |/ OPA
*             _|_               _|_
*             ///               /// 

.SUBCKT OPA   INP    INN    OUT



EGN1        O1     0      INP    INN    -1
EGN2        OUT    0      IP     0      -1000000
R1          O1     IP     10k 
C1          OUT    IP     10p
.ends

.SUBCKT ONE_S IN      OUT 
EGN1        O1       0      IP     0      -1000000
EGN2        OUT      0      O1     0      -1
R1          IN       IP     1 
C1          IP       O1     1
.ends

.end

=====================END_OF_SPICE=======================

Q is control by the ratio of R2 to the other resistors.
For a Q=10 there is a peaking of 20dB.

For a Q=1 there is little peaking.




